
*=============Free_Running_Serial_CRYSTAL_OSC============
How well does Spice simulate a free running oscillator?
Being able to use the output waveform data to see time delays 
can come in handy.

*=========CRYSTAL_SubCircuit============================================
.SUBCKT   XCRYST VA     VB
 RS       VA     VCS    340
 CS       VCS    VLS    7f
 LS       VLS    VB     3.5
 CP       VA     VB     .03pf
.ENDS     XCRYST
*                             _  _  _
*    ___             VCS  VLS/ \/ \/ \    ___
*   |VA |__/\  /\  /\__||____| () () |___|VB |
*   |___| |  \/  \/    ||              | |___|
*         | RS         CS     LS       |  _|_
*         |                            |  ///
*         |____||______________________|
*              || CP
*                                   CS LS = 29.97meg @ 885Gohms 
*                      ___
*              ___   ||   ||   ___
*             |VA |__||   ||__|VB |   
*             |___|  ||   ||  |___|
*                    ||___||
*
* Freq                 Mode  L1     C1       R1    C0      Q
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
===========================================================================
One can make a sub-circuit using typical crystal models.
In this case a 1MegHz crystal is being used to run 
in the series mode. 

*=========CRYSTAL_Oscillator============================================
*
*    ^  VCC                _________
*   /_\                   |         |
*   _|_  D1               |   |\    |   ____
*    ^         R1         |___| \/\_|__|VREF|
*   /_\   1Meg                | /\/    |____|
*    |__/\  /\  /\__          |/
*    |    \/  \/    |           XINV4
*    |              |  C1
*    |     |\       |      |\
*    |_____| \/\____|__||__| \/\__
*    |     | /\/       ||  | /\/  |
*    |     |/              |/     |
*    |       XINV1          XINV2 |
*    |      ___                   |
*   _|_   ||   ||   ___           |
*  |VA |__||   ||__|OUT|__________|
*  |___|  ||   ||  |___|
*   _|_   ||___||
*    ^          XCRY1
*   /_\  D2
*   _|_
*   ///
*
===========================================================================
In this mode, the Crystal is operating more like a short circuit.
The phase across the crystal is expected to be zero.  

*=========Run_transient============================================
*.OPTIONS GMIN=1e-18    METHOD=trap   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1  
VCC       VCC    0      DC      5
XCRY1     VA     OUT    XCRYST
XINV      VA     VC     VCC     INV_R
XINV2     VD     OUT    VCC     INV_R
XINV4     VREF   VREF   VCC     INV_R
R1        VA     VC     1meg
C1        VC     VD     .1u    IC=0
D1        VA     VCC    DD
D2        0      VA     DD
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran     10n    600u  0     10n    UIC

.control
run
set       pensize = 2
plot      va    out  vref



plot      out   va   vref
plot      out   vc   va  vref xlimit 590u 600u 
print     mean(vref) 
===========================================================================
There is always a question of how long does it
take to reach steady state.

===========================================================================
One needs to look at the last few cycles over 
a smaller time to see some detail. 

===========================================================================
Some of the nonlinearities are visible. Notice that
some clamp diodes have been added to simulate ESD diodes.

*
*    ^  VCC                _________
*   /_\                   |         |
*   _|_  D1               |   |\    |   ____
*    ^         R1         |___| \/\_|__|VREF|
*   /_\   1Meg                | /\/    |____|
*    |__/\  /\  /\__          |/
*    |    \/  \/    |           XINV4
*    |              |  C1
*    |     |\    VC |      |\
*    |_____| \/\____|__||__| \/\__
*    |     | /\/       ||  | /\/  |
*    |     |/            VD|/     |
*    |       XINV1          XINV2 |
*    |      ___                   |
*   _|_   ||   ||   ___           |
*  |VA |__||   ||__|OUT|__________|
*  |___|  ||   ||  |___|
*   _|_   ||___||



*    ^          XCRY1
*   /_\  D2
*   _|_
*   ///
*
===========================================================================
But how well can the timing be trusted? One needs 
a mathematical way to process the output data. 
In this example the last 50 transitions will be looked
at. 

*=========Process_Output_Arrays========
let       Numb_Pt= length(out)
compose   time_tp start = 0 stop = 49 step =1
compose   dt_tp   start = 0 stop = 49 step =1
*=========Look_at_Last_50_edges========
let       i = Numb_Pt-2
let       n = 0
while     (n < 50)
if        ( out[i]-2.5 >= 0 & out[i-1] -2.5 <= 0)
let       tp = time[i]
let       time_tp[n]=tp
* echo    $&n   $&tp    
let       n=n+1
endif
let i=i-1
endwhile
===========================================================================
The oscillator's period can be found by subtracting the
edge time points

*=========Find_and_Plot_periods========
let       i = 0
repeat    49
let       dt_tp[i] = -time_tp[i+1]+time_tp[i]  
let       i=i+1
endrepeat
let       dt_tp[49] =dt_tp[48]
plot      dt_tp vs time_tp
===========================================================================
And plotting the time periods versus time can show
all the simulation errors. 

*=========Print_Frequency========
print     mean(1/dt_tp)
.endc
======================================================================
The mean can be taken and the output frequency can 
be printed out.

Circuit:         OneMeg_FreeRun_Serial_Crystal
mean(vref) =     2.42844e+00
mean(1/dt_tp) =  1.01649e+06

======================================================================
This value should come close to what an AC analysis 
will predict for the sub-circuit below.



*=========CRYSTAL_SubCircuit============================================
.SUBCKT   XCRYST VA     VB
 RS       VA     VCS    340
 CS       VCS    VLS    7f
 LS       VLS    VB     3.5
 CP       VA     VB     .03pf
.ENDS     XCRYST
*                             _  _  _
*    ___             VCS  VLS/ \/ \/ \    ___
*   |VA |__/\  /\  /\__||____| () () |___|VB |
*   |___| |  \/  \/    ||              | |___|
*         | RS         CS     LS       |  _|_
*         |                            |  ///
*         |____||______________________|
*              || CP
*                                   CS LS = 29.97meg @ 885Gohms 
*                      ___
*              ___   ||   ||   ___
*             |VA |__||   ||__|VB |   
*             |___|  ||   ||  |___|
*                    ||___||
*
* Freq                 Mode  L1     C1       R1    C0      Q
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
===========================================================================

===========Full_Netlist_For_Copy_Paste=======================
OneMeg_FreeRun_Serial_Crystal
*
*    ^  VCC                _________
*   /_\                   |         |
*   _|_  D1               |   |\    |   ____
*    ^         R1         |___| \/\_|__|VREF|
*   /_\   1Meg                | /\/    |____|
*    |__/\  /\  /\__          |/
*    |    \/  \/    |           XINV4
*    |              |  C1
*    |     |\       |      |\
*    |_____| \/\____|__||__| \/\__
*    |     | /\/       ||  | /\/  |
*    |     |/              |/     |
*    |       XINV1          XINV2 |
*    |      ___                   |
*   _|_   ||   ||   ___           |
*  |VA |__||   ||__|OUT|__________|
*  |___|  ||   ||  |___|
*   _|_   ||___||
*    ^          XCRY1
*   /_\  D2
*   _|_
*   ///
*
* .OPTIONS  GMIN=1e-18    METHOD=trap   ABSTOL=1e-18  TEMP=27 srcsteps = 1  gminsteps = 1  
VCC       VCC    0      DC      5
XCRY1     VA     OUT    XCRYST
XINV      VA     VC     VCC     INV_R
XINV2     VD     OUT    VCC     INV_R
XINV4     VREF   VREF   VCC     INV_R
R1        VA     VC     1meg
C1        VC     VD     .1u    IC=0
D1        VA     VCC    DD
D2        0      VA     DD

*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
.tran     10n    600u  0     10n    UIC

.control
run
set       pensize = 2
plot      va    out  vref
plot      out   va   vref
plot      out   vc   va  vref xlimit 590u 600u 
print     mean(vref) 

*=========Process_Output_Arrays========
let       Numb_Pt= length(out)
compose   time_tp start = 0 stop = 49 step =1
compose   dt_tp   start = 0 stop = 49 step =1

*=========Look_at_Last_50_edges========

let       i = Numb_Pt-2
let       n = 0



while     (n < 50)
if        ( out[i]-2.5 >= 0 & out[i-1] -2.5 <= 0)
let       tp = time[i]
let       time_tp[n]=tp
* echo    $&n   $&tp    
let       n=n+1
endif
let i=i-1
endwhile

*=========Find_and_Plot_periods========
let       i = 0
repeat    49
let       dt_tp[i] = -time_tp[i+1]+time_tp[i]  
let       i=i+1
endrepeat
let       dt_tp[49] =dt_tp[48]
plot      dt_tp vs time_tp

*=========Print_Frequency========
print     mean(1/dt_tp)
.endc

.model    DD            D(IS=3.15e-18 RS=100 CJO=10f ) 

*=========CRYSTAL_SubCircuit============================================
.SUBCKT   XCRYST VA     VB
 RS       VA     VCS    340
 CS       VCS    VLS    7f
 LS       VLS    VB     3.5
 CP       VA     VB     .03pf
.ENDS     XCRYST
*                             _  _  _
*    ___             VCS  VLS/ \/ \/ \    ___
*   |VA |__/\  /\  /\__||____| () () |___|VB |
*   |___| |  \/  \/    ||              | |___|
*         | RS         CS     LS       |  _|_
*         |                            |  ///
*         |____||______________________|
*              || CP
*                                   CS LS = 29.97meg @ 885Gohms 
*                      ___
*              ___   ||   ||   ___
*             |VA |__||   ||__|VB |   
*             |___|  ||   ||  |___|
*                    ||___||
*
* Freq                 Mode  L1     C1       R1    C0      Q
* 1 MHz                fund  3.5H   0.007pf  340   3pf     64679
*
*=========Real_Inverter=========================
.SUBCKT   INV_R  IN    OUT    VCC
 MN1      OUT    IN    0      0      NMOSC   W=30u     L=1u 
 MP1      OUT    IN    VCC    VCC    PMOSC   W=90u     L=1u 
 C1       OUT    0     1p
.ENDS     INV_R
*
*               ^
*  INV_R       /_\ VCC
*               |
*             <-
*           ||___   480ps at 6 stages
*         __||
*  ___   |  ||_ MP1
* |VIN|__|     |   ___
* |___|  |     |__|VOUT
*        |    _|  |___|
*        |__||
*           ||___ MN1
*           ||->
*             _|_
*             ///
*
*==================CMOS_Models=========================
.model             NMOSC              NMOS(
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=160e-10      xj=0.25e-06        nch=0.5e+17                        
*------------------V_threshold------------------------------------------- 
+ vth0=0.72        nlx=0.12e-06                                          
*------------------Bulk-------------------------------------------------- 
+ k1=1.04          k2=-1.209E-01
+ cdsc=-2.4E-4     cdscd=-1.506E-04   cdscb=-2.219E-04                   
*------------------mobility----------------------------------------------



+ u0=678           ua=8.964e-10                                          
+ ub=1.472e-18     uc=-4.441E-17      vsat=86000                         
*------------------Subthresshold-----------------------------------------
+ nfactor=1.8                                                               
+ cit=-5.0E-04     voff=-7.862E-02                                        
+ eta0=4.441e-16   etab=-2.E-01       dsub=0.7                           
*------------------Hot electrons----------------------------------------- 
* alpha0=1.61e-05  beta0=36.68                                           
*------------------VAF---------------------------------------------------
+ lint=.12e-06     pclm=.19           pscbe1=3.79e+08   pscbe2=9.4e-05   
+ delta=0.01655    pvag=0.4484                                            
*------------------Bulk_diode-------------------------------------------- 
+ js=5.858e-08                                                              
*------------------Resistance-------------------------------------------- 
+ rsh=70           rdsw=375                                                
+ wr=0.7586        prwb=0             prwg=-4.441E-17                    
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873    
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10                         
+ pb=0.5614        pbsw=0.8           xpart=0                             
+ dlc=5e-08        dwc=1.5e-07  
*------------------BulkChargeEffect--------------------------------------  
* a0=0.7           a1=0                a2=1                ags=0.05583    
* b0=6.305e-08     b1=6.579e-08        keta=-1.531E-02                    
*------------------ShortChannel------------------------------------------ 
+ dvt0=2.2         dvt1=0.53          dvt2=-1.521E-01      drout=0.76       
+ pdiblcb=.4       pdiblc1=0.00886    pdiblc2=0.00029                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                          
+ ww=-9.525E-14    wwn=1.0                                               
+ dvt0w=0          dvt1w=5.3e6        dvt2w=-1.E-01                        
+ k3=2.53          k3b=-5             dwg=0                dwb=0  
*------------------Noise-------------------------------------------------
* af=1             kf=1e-28           ef=0.95                            
*------------------Temperature------------------------------------------- 
* pvsat=0          ute=-1.258E+00     kt1=-3.85E-01                      
* kt1l=0           kt2=-3.098E-02     ua1=5.705e-09                      
* ub1=-1.147E-17   uc1=-1.302E-01     at=20380                           
* prt=-3.287E+02   lk1=0              lk2=0                              
* lvsat=0          la0=0              lags=0             lute=0          
+ luc=0                                                                )      
                             
.model             PMOSC              PMOS(
+ Level=  8        Tnom=27.0
*------------------Process----------------------------------------------- 
+ tox=1.725e-08    xj=2e-07           nch=1e+17                         
*------------------V_threshold------------------------------------------- 
+ vth0=-0.90       nlx=1.84e-08                                          
*------------------Bulk-------------------------------------------------- 
+ k1=0.3969        k2=0.03536                                           
*------------------mobility----------------------------------------------
+ u0=268           ua=4.124e-09                                          
* ub=-3.8E-19      uc=-1.512E-02      vsat=140000                        
*------------------Subthresshold-----------------------------------------
+ nfactor=.5016                                                             
* cit=0.0001       voff=-0.08          nfactor=1.0016   vth0=-0.94        
* eta0=0.038       etab=-1.057E-02     dsub=0.3501                       
*------------------Hot electrons----------------------------------------- 
+ alpha0=2.5e-08     beta0=28.92                                           
*------------------VAF---------------------------------------------------
+ lint=.1e-06      pclm=3.2            pscbe1=5e+08       pscbe2=1e-05   
* delta=0.009      pvag=4                                                
*------------------Bulk_diode-------------------------------------------- 
+ js=3.95e-08                                                              
*------------------Resistance-------------------------------------------- 
+ rsh=135          rdsw=2500                                               
* prwg=-7.2E-02    prwb=0.06          wr=0.8625                          
*------------------Capacitance-------------------------------------------
+ cj=0.0002424     cjsw=2.73e-10      mj=0.3551           mjsw=0.3873    
+ cgso=9e-13       cgdo=9e-13         cgbo=7e-10                         
* cdsc=-2.4E-4     cdscd=-1.506E-01   cdscb=-2.219E-04                   
* pb=0.85          pbsw=0.88          xpart=0                             
* dlc=5e-08        dwc=1.5e-07                                           
*------------------BulkChargeEffect--------------------------------------  
* a0=1             a1=0               a2=2                 ags=0.268      
* b0=5e-07         b1=1e-07           keta=-1.0E-02                       
*------------------ShortChannel------------------------------------------
* dvt0=2.9         dvt1=0.2           dvt2=-1.521E-01      drout=0.175   
* pdiblcb=0        pdiblc1-           pdiblc2=0.001                      
*------------------NarrowChannel-----------------------------------------
+ w0=2.6e-04       wint=0.16e-06                                         
+ ww=-3.1E-19      wwn=1.9                                               
* dvt0w=0.68       dvt1w=5.3e6        dvt2w=0.051                          
* k3=56            k3b=-3             dwg=0               dwb=1e-08       



*------------------Noise-------------------------------------------------       
+ af=0.8           kf=1.50e-30        ef=0.95                            
*------------------Temperature------------------------------------------- 
* ldelta=0.02282   lpdiblc1=0.01877   ute=-1.500E+00                       
* cgsl=1.5e-10     cgdl=1.5e-10       ckappa=0.2463      cf=0            
* kt1=-4.684E-01   kt1l=-2.0E-08      kt2=-2.818E-02                     
* ua1=-2.E-10      ub1=-4.5E-18       uc1=-2.000E-02                     
+ at=-1.5E+05      prt=1400                                             )

.end

